


1 Complex PK model

In this example,

- one type of dose is administered orally ($\text{type}=1$) and absorbed into a latent compartment following a first-order absorption process,
- a second type is administered orally ($\text{type}=2$) and absorbed into the central compartment following a zero-order absorption process,
- a third type is directly administered intravenously to the central compartment ($\text{type}=3$).

The transfer from the latent to the central compartment is linear. A peripheral compartment is linked to the central compartment. The drug is eliminated by a linear process from the latent compartment and a nonlinear process from the central one. Here, A_l and A_c are the drug amounts in the latent and central compartments.



Implementation of this model using PK macros:

```

[LONGITUDINAL]
input = {F1, F2, ka, Tk0, k1, k23, k32, V, k, Vm, Km}


PK:
compartment(cmt=1, amount=A1)
compartment(cmt=2, amount=Ac)
peripheral(k23,k32)
oral(type=1, cmt=1, ka, p=F1)
oral(type=2, cmt=2, Tk0, p=F2)
iv(type=3, cmt=2)
transfer(from=1, to=2, kt=k1)
elimination(cmt=1, k)
elimination(cmt=2, Km, Vm)
Cc = Ac/V

```

pk_model.txt



Example of R script for computing the concentration in the central compartment using `simulx`:



2 Hierarchical model

Mixed effects models are hierarchical models that involves different types of variables:

- We call $y_i = (y_{ij}, 1 \leq j \leq n_i)$ the set of *longitudinal data* recorded at times $(t_{ij}, 1 \leq j \leq n_i)$ for subject i , and \mathbf{y} the combined set of observations for all N individuals: $\mathbf{y} = (y_1, \dots, y_N)$.
- We write ψ_i for the vector of *individual parameters* for individual i and $\boldsymbol{\psi}$ the set of individual parameters for all N individuals: $\boldsymbol{\psi} = (\psi_1, \dots, \psi_N)$.
- The distribution of the individual parameters ψ_i of subject i may depend on a vector of *individual covariates* c_i .
- In a population approach context, we call θ the vector of *population parameters*.

Considering these variables as random variables, the joint probability distribution of \mathbf{y} , $\boldsymbol{\psi}$, \mathbf{c} and θ can be decomposed into a product of conditional distributions:

$$p(\mathbf{y}, \boldsymbol{\psi}, \mathbf{c}, \theta) = p(\mathbf{y}|\boldsymbol{\psi}, \theta)p(\boldsymbol{\psi}|\mathbf{c}, \theta)p(\mathbf{c}|\theta)p(\theta)$$

MLXTRAN takes advantage of the hierarchical structure of this joint probability distribution by decomposing the joint model into several submodels.

The example below shows how each component of the model is implemented in a different section. Each section combines equations and/or definitions.

```

hierarchical_model.txt
[LONGITUDINAL]
input = {V, k, b}
EQUATION:
D=100
f = D/V*exp(-k*t)
DEFINITION:
y = {distribution=normal, prediction=f, sd=b*f}


[INDIVIDUAL]
input = {V_pop, omega_V, w, w_pop}
EQUATION:
V_pred = V_pop*(w/w_pop)
DEFINITION:
V = {distribution=logNormal, prediction=V_pred, sd=omega_V}

[COVARIATE]
input = {w_pop, omega_w}
DEFINITION:
w = {distribution=normal, mean=w_pop, sd=omega_w}

[POPULATION]
input = {ws, gw, Vs, gV}
DEFINITION:
w_pop = {distribution=normal, mean=ws, sd=gw}
V_pop = {distribution=logNormal, mean=log(Vs), sd=gV}


```

The `simulx` script now includes the values of the *hyperparameters* that define the distribution of the population parameters. In this example, 2 populations and 3 individuals in each population with their own covariates are simulated:



3 Multiple outputs model

Joint model for PK, continuous PD and categorical PD data:

```
joint_model.txt   
  
[LONGITUDINAL]  
input = {ka, V, k, a1, a2, Emax, EC50, th1, th2, th3}  
  
PK:  
depot(target=Ad)  
  
EQUATION:  
ddt_Ad = -ka*Ad  
ddt_Ac = ka*Ad - k*Ac  
Cc = Ac/V  
E = Emax*Cc/(EC50 + Cc)  
lp1 = th1 + th2*Cc  
lp2 = th1 + th2*Cc + th3  
  
DEFINITION:  
Concentration = {distribution=lognormal, prediction=Cc, sd=a1}  
  
Effect = {distribution=normal, prediction=E, sd=a2}  
  
Level = {type=categorical, categories={1,2,3},  
          logit(P(Level<=1)) = lp1  
          logit(P(Level<=2)) = lp2 }
```


We use `simulx` for computing quantities and simulating data from this model.




4 Time-to-event data model

The distribution of time-to-event data (or *survival* data) is defined with the *hazard* function. Some additional information should be provided in the model (maximum number of events, right censoring time, interval censoring, ...).


We consider a Weibull model for a single event in this example, with a right censoring time equal to 100.

```
event1_model.txt   
  
[LONGITUDINAL]  
input = {beta,lambda}  
  
EQUATION:  
h=(beta/lambda)*(t/lambda)^(beta-1)  
  
DEFINITION:  
e = {type=event, maxEventNumber=1, rightCensoringTime=100, hazard=h}
```

The following model uses the same hazard function, assuming now that the events are repeated and interval censored:

```
event2_model.txt   
  
[LONGITUDINAL]  
input = {beta,lambda}  
  
EQUATION:  
h=(beta/lambda)*(t/lambda)^(beta-1)  
  
DEFINITION:  
e = {type=event, eventType=intervalCensored,  
      intervalLength=10, rightCensoringTime=100, hazard=h}
```

We can use `simulx` for simulating 100 individuals with these two models. It is mandatory to state explicitly when the experiment starts (at time $t = 0$ in this example):

```
event1_simulx.R   
  
p <- list( name = c('beta','lambda'), value=c(2.5,50))  
h <- list(name='h', time=seq(from=0, to=30, by=0.1))  
e <- list(name='e', time=0)  
o <- list(h, e);  
g <- list(size=100)  
res1 <- simulx(model='event1_model.txt',parameter=p,output=o,group=g)  
res2 <- simulx(model='event2_model.txt',parameter=p,output=o,group=g)
```